

cy 001

NAVAL UNDERWATER SYSTEMS CENTER
NEW LONDON LABORATORY
NEW LONDON, CT 06320

TECHNICAL MEMORANDUM

Automatic Directivity Index Calculation Program

Date: 23 August 1984

Prepared by:

John D. Rookard
JEHU D. ROOKARD

Submarine Sonar Department
Transducers and Arrays Division
Electroacoustic Transduction Branch

Approved for Public Release; distribution unlimited

Report Documentation Page				Form Approved OMB No. 0704-0188	
Public reporting burden for the collection of information is estimated to average 1 hour per response, including the time for reviewing instructions, searching existing data sources, gathering and maintaining the data needed, and completing and reviewing the collection of information. Send comments regarding this burden estimate or any other aspect of this collection of information, including suggestions for reducing this burden, to Washington Headquarters Services, Directorate for Information Operations and Reports, 1215 Jefferson Davis Highway, Suite 1204, Arlington VA 22202-4302. Respondents should be aware that notwithstanding any other provision of law, no person shall be subject to a penalty for failing to comply with a collection of information if it does not display a currently valid OMB control number.					
1. REPORT DATE 23 AUG 1984		2. REPORT TYPE Technical Memo		3. DATES COVERED 23-08-1984 to 23-08-1984	
4. TITLE AND SUBTITLE Automatic Directivity Index Calculation Program				5a. CONTRACT NUMBER	
				5b. GRANT NUMBER	
				5c. PROGRAM ELEMENT NUMBER	
6. AUTHOR(S) Jehu Rookard				5d. PROJECT NUMBER	
				5e. TASK NUMBER	
				5f. WORK UNIT NUMBER	
7. PERFORMING ORGANIZATION NAME(S) AND ADDRESS(ES) Naval Underwater Systems Center, New London, CT, 06320				8. PERFORMING ORGANIZATION REPORT NUMBER TM 841152	
9. SPONSORING/MONITORING AGENCY NAME(S) AND ADDRESS(ES)				10. SPONSOR/MONITOR'S ACRONYM(S)	
				11. SPONSOR/MONITOR'S REPORT NUMBER(S)	
12. DISTRIBUTION/AVAILABILITY STATEMENT Approved for public release; distribution unlimited					
13. SUPPLEMENTARY NOTES NUWC2015					
14. ABSTRACT A computer program was written for the HP 9845 and the HP 9816 to calculate the directivity index directly from the beam pattern. A graphics tablet and stylus is used to digitize data points used to compute the directivity index. Graphical output is produced on a printer and plotter.					
15. SUBJECT TERMS directivity index					
16. SECURITY CLASSIFICATION OF:			17. LIMITATION OF ABSTRACT Same as Report (SAR)	18. NUMBER OF PAGES 27	19a. NAME OF RESPONSIBLE PERSON
a. REPORT unclassified	b. ABSTRACT unclassified	c. THIS PAGE unclassified			

ABSTRACT

A computer program was written for the HP 9845 and the HP 9816 to calculate the directivity index directly from the beam pattern. A graphics tablet and stylus is used to digitize data points used to compute the directivity index. Graphical output is produced on a printer and plotter.

INTRODUCTION

This program was developed to enable the user to digitally input beam pattern data points into a computer and calculate the directivity index from this data. It eliminates the slow and monotonous task of visually determining the polar coordinates of the data points and manually keying them into the computer for processing.

Prior to the creation of this program, two methods were used to find the directivity index of a beam pattern. The first method involved extracting data from the beam pattern by reading a data point at specific, regular intervals. This method was slow and less accurate due to the fact that the operator introduced error when approximating the position of the data points. The second method required the beam pattern, which was plotted in a polar coordinate system, to be re-plotted in a rectangular coordinate system. This operation was performed by the user. Once this conversion had been completed, a mechanical device called a planimeter was used to calculate the directivity index. Obviously, this method was extremely slow. Therefore, this program greatly reduces the time and effort required to calculate the directivity index of a beam pattern. It also produces a more accurate result.

The program requires an HP 9845 computer, an HP 9111A graphics tablet and an HP 9872A plotter. In addition, it requires that the Structured Programming ROM be resident in the HP 9845 computer. Without it, certain BASIC statements cannot be executed.

A slightly modified version of this program is available for the HP 9816 computer. It provides the same capabilities as the HP 9845 version but utilizes an HP 7470 plotter instead of an HP 9872A plotter. The HP 9816 version also requires that the BASIC extensions version 2.1 be present.

DESCRIPTION OF PROGRAM

The "CALCDI" program is responsible for converting a polar plot representing a beam pattern data into a finite number of polar coordinate data points. It is from these data points that the directivity index is computed.

The program is divided into subprograms which perform the functions of initializing the graphics tablet, digitizing a directivity pattern, calculating the directivity index, and producing a hard copy output. Subprograms are also employed to perform conversions between rectangular and polar coordinate systems and perform functions not found in HP BASIC. These programming techniques result in a program that is easy to understand and modify. Subprogram modules can also be used to create new programs.

PROBLEM DEFINITION AND SOLUTION

As stated in the introduction, this program's purpose is to allow the operator to digitize a finite number of data points from a directional response pattern. The number of data points collected is dependent upon the incremental angle specified by the operator. After collecting the data, the beam pattern's directivity index calculation is performed and the result is displayed. The beam pattern is plotted on the CRT to allow the operator to verify that no errors have occurred during the data entry process.

Looking at the program from an overall view, it can be divided into 3 distinct steps. The first step involves digitizing a beam pattern using the graphics tablet. This requires converting the rectangular coordinates from the graphics tablet to the polar coordinates required by the directivity index formula. The second step is to use the polar coordinate data points to compute the directivity index. This entails changing the formula for calculating the directivity index into a procedure able to be executed by the computer. The directivity index formula, an integral, is translated into a summation. Since the original rectangular coordinate data points are not saved, the polar coordinate data points must be converted back to the rectangular system to enable them to be displayed on the CRT, printer or plotter.

STEP 1

DIGITIZING A DIRECTIONAL RESPONSE PATTERN

A beam, or directional response, pattern is digitized by placing it on the surface of the graphics tablet and tracing it with the stylus. Although the beam pattern is printed on polar coordinate paper, X and Y coordinates are entered into the computer as the digitizing process occurs. These X and Y coordinates must be converted to polar form for use by the directivity index formula.

Polar coordinate points are in the form (Rho, Theta) with Rho equalling the distance from the origin and Theta equalling an angle in the range 0-360 degrees. Rho and Theta must be derived from the X and Y coordinates. This conversion process is accomplished by the following subprograms: "Displacement," "Calculateradius," "Calcarctangent," "Determinequad," and "Convertto360." Each of these subprograms is well commented. The procedure for the conversion is also presented in algorithmic form.

STEP 2

CALCULATE THE DIRECTIVITY INDEX

The directivity index calculation is the most important segment of the program. From the polar coordinate data, we are able to determine the directivity index. This computation is performed by the subprogram called "Calcdirindex."

The directivity index is equal to ten times the common logarithm of the directivity factor. The directivity factor is defined as the ratio of the intensity in a reference direction, usually the axis, to the intensity average over all directions.

The directivity factor is calculated using the formula as presented by Bobber in Underwater Electroacoustic Measurements (Washington D.C., NRL, 1970, p. 84)

From Bobber,

$$\text{directivity factor} = \frac{2}{\Delta \theta \cdot \sum_{i=0}^n A_i} \quad (1)$$

where $\Delta \theta$ equals the degrees resolution per angular interval, n equals the number of sampling points per 180 degrees, and

$$A_i = \left[\frac{p(\theta_i)}{p_0} \right]^2 \cdot \sin \theta_i \quad (2)$$

In equation 2, θ_i is the angular measure in degrees of the polar coordinate data point. Also,

$$\left[\frac{p(\theta_i)}{p_0} \right]^2 = 10^{-(\text{db down}/10)} \quad (3)$$

where db down equals the difference of the maximum response axis and the radius of the polar coordinate data point. The formula for sampling the directivity pattern through 360 degrees becomes,

$$\text{directivity factor} = \frac{4}{\Delta \theta \sum_{i=0}^{2n} A_i} \quad (4)$$

Consider the case of $\Delta \theta = 5$ degrees, the default value for the resolution in "Calcdirindex." Converting $\Delta \theta$ to radians,

$$\text{directivity factor} = \frac{4}{\Delta \theta \cdot \frac{\pi}{180} \sum_{i=0}^{72} |A_i|} \quad (5)$$

Simplifying,

$$\text{directivity factor} = \frac{229.2}{\Delta \theta \cdot \sum_{i=0}^{72} |A_i|} \quad (6)$$

The directivity factor formula of Equation 6 is transformed into a sequence of BASIC statements. "Calcdirindex" simulates the summation process by utilizing an iterative loop. The variable "resolution" corresponds to $\Delta \theta$ and 229.2 is a numeric constant.

If a directivity pattern has symmetry about the equator of the measuring sphere, then it is only necessary to use one half of the pattern in the computation. However, some patterns are nearly rotationally symmetric, but not exactly. For these "near symmetric patterns," it is desirable to average the two directivity factors as calculated for the two halves of the pattern. Therefore, it must be noted that this program assumes all beam patterns are nearly rotationally symmetric and thus performs the summation over the full 360 degrees.

STEP 3

DISPLAY THE DIRECTIVITY PATTERN ON THE OUTPUT DEVICE


The polar coordinate data is converted to X and Y coordinates so that it may be output to the CRT, printer, and plotter. This is accomplished by the subprograms called "Displaypattern," "Printcopy," and "Plotcopy."

Pseudo-code

Main Program

1. Clear the CRT
 - a. Clear the alphanumeric raster
 - b. Clear the graphics raster
 2. Print Identifying message
 3. Wait for person to read message
 - a. Continue when operator hits "Continue" key
 4. Digitize the directivity pattern using the 9111A Graphics Tablet
 5. Calculate the directivity index from the polar coordinate values entered from the graphics tablet
 6. Draw the directivity pattern on the CRT and print the directivity index
 7. Output directivity pattern to printer with option of suppressing output
 8. Output directivity pattern to plotter with option of suppressing output
 9. Clear the CRT
 - a. Clear the alphanumeric raster
 - b. Clear graphics raster
 10. See if operator has more directivity patterns to digitize
 - If choice = yes then
 - Repeat procedure from step 4
 - Else
 - End of program
- End if

MAIN PROGRAM FLOWCHART



Pseudo Code

Algorithm to Convert from the Rectangular
to the Polar Coordinate System

1. Input the X and Y coordinates
2. Determine its X and Y displacements from the origin in user units
3. Calculate Rho (Radius) using the X and Y displacements with Rho being measured in user units
4. Compute the principle angle given the X and Y displacements from the origin
5. Determine which quadrant the point lies in based upon its relative position to the origin
6. Given the principle angle and the quadrant, produce an angle in the range 0-360 degrees

VARIABLES

- originx - the x-coordinate of the origin of the directivity pattern in digitizing units.
- originy - the y-coordinate of the origin of the directivity pattern in digitizing units.
- zerodegfsx - the x-coordinate of the point representing zero degrees full scale on the directivity pattern. It is measured in digitizing units.
- zerodegfsy - the y-coordinate of the point representing zero degrees full scale on the directivity pattern. It is measured in digitizing units.
- mmax - the x-coordinate of the point representing the maximum response axis. It is measured in digitizing units.
- mmaxy - the y-coordinate of the point representing the maximum response axis. It is measured in digitizing units.
- gtabuperdb - number of digitizing units (graphic tablet units) per decibel. This is used as a scaling or conversion factor.
- dbfullscale - number of decibels from the origin to zero degrees full scale.
- radiusvalues (*) - an array representing the radii of the polar coordinate data points.
- angle (*) - an array representing the angles of the polar coordinate data points.
- mra - the maximum response axis value.
- resolution - the incremental angle, which determines the number of data points gathered.
- dirindex - the value of the directivity index.
- skew - an offset angle to compensate for directivity patterns positioned crookedly on the graphics tablet digitizing area.

```

10  ! *****
20  ! *
30  ! *          AUTOMATIC DIRECTIVITY INDEX CALCULATION PROGRAM
40  ! *          by JEHU ROOKARD
50  ! *
60  ! *          Created 2 AUG 1984
70  ! *
80  ! *****
90  !
100 ! THIS PROGRAM DIGITIZES A DIRECTIVITY PATTERN AND CALCULATES THE
110 ! DIRECTIVITY INDEX
120 !
130 !
140 Main:      ! MAIN DIRECTIVITY PATTERN PROGRAM
150  OPTION BASE 0
160  COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mmax,Mmay
170  COM Gtabuperdb,Dbfullscale
180  COM Radiusvalues(360),Angle(360)
190  COM Mra
200  COM Resolution
210  COM Dirindex
220  COM Skew
230  CALL Clearscreen
240  PLOTTER IS 13,"GRAPHICS"      ! ACTIVATE CRT GRAPHICS RASTER
250  GRAPHICS
260  GCLEAR                      ! CLEAR THE GRAPHICS RASTER
270  PLOTTER 13 IS OFF           ! DEACTIVATE CRT GRAPHICS RASTER
280  PRINT "THIS PROGRAM ENTERS A DIRECTIVITY PATTERN FROM THE GRAPHICS"
290  PRINT "TABLET AND CALCULATES THE DIRECTIVITY INDEX"
300  INPUT "PRESS 'CONT' WHEN READY TO PROCEED",Cont$
310  CALL Enterapattern          ! ENTER A DIRECTIVITY PATTERN USING THE 9111A
320                                ! GRAPHICS TABLET
330  CALL Calcdirindex           ! CALCULATE THE DIRECTIVITY INDEX
340  CALL Displaypattern         ! DRAW THE DIRECTIVITY PATTERN ON THE CRT
350  CALL Printcopy             ! MAKE HARD COPY ON PRINTER
360  CALL Plotcopy              ! MAKE HARD COPY ON PLOTTER
370  PLOTTER IS 13,"GRAPHICS"    ! CLEAR THE GRAPHICS RASTER
380  GCLEAR
390  EXIT GRAPHICS
400  PLOTTER 13 IS OFF           ! DEACTIVATE CRT GRAPHICS RASTER
410  CALL Clearscreen
420  Yesno$="Y"
430  INPUT "ENTER ANOTHER PATTERN - Y/N. DEFAULT IS 'Y'",Yesno$
440  IF Yesno$="Y" THEN
450  PLOTTER IS 13,"GRAPHICS"    ! CLEAR THE GRAPHICS RASTER
460  GCLEAR
470  PLOTTER 13 IS OFF           ! DEACTIVATE THE CRT GRAPHICS RASTER
480  GOTO 310
490  ELSE
500  END
510  !
520  !
530  !
540 Enterapattern:      ! PROCEDURE TO ENTER A DIRECTIVITY PATTERN FROM THE 9111A
550                      ! GRAPHICS TABLET
560  SUB Enterapattern
570  CALL Clearscreen          ! CLEAR THE ALPHA RASTER
580  PLOTTER IS 7,6,"9872A"    ! THE 9111A GRAPHICS TABLET CAN ONLY BE ACCESSED
590                                ! THROUGH THE GRAPHICS ROM. ALTHOUGH IT IS A
600                                ! GRAPHICS TABLET, THE COMPUTER TREATS IT AS A
610                                ! PLOTTER
620  OUTPUT 706;"IN"          ! INITIALIZE THE 9111A GRAPHICS TABLET
630  CALL Setupaxes           ! GET THE NECESSARY POINTS TO PROPERLY SCALE
640                                ! THE DIGITIZING AREA AND ALL POINTS ENTERED
650                                ! FROM IT

```

```

660 CALL Entercurve          ! DIGITIZE THE ACTUAL DIRECTIVITY PATTERN
670 SUBEND
680 !
690 !
700 !
710 Setupaxes:              ! ENTER THE NECESSARY SCALING VALUES FROM THE GRAPHICS
720                          ! TABLE SO THAT A BASIS FOR ENTERING ALL SUCCEEDING
730                          ! POINTS IS ESTABLISHED
740                          ! A METHOD (SCALE) FOR CONVERTING BETWEEN DIGITIZING UNITS
750                          ! AND USER UNITS IS SET UP
760 SUB Setupaxes
770 OPTION BASE 0
780 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mrax,Mray
790 COM Gtabuperdb,Dbfullscale
800 COM Radiusvalues(360),Angle(360)
810 COM Mra
820 COM Resolution
830 COM Dirindex
840 COM Skew
850 Dbfullscale=50
860 INPUT "ENTER NO. OF db'S FROM ORIGIN TO ZERO DEGREES FULL SCALE. DEFAULT I
S 50",Dbfullscale
870 IF Dbfullscale>50 THEN
880 PRINT "NO VALUES GREATER THAN 50 ALLOWED. PLEASE RE-ENTER"
890 GOTO 850
900 ELSE
910 END IF
920 !
930 Resolution=5
940 INPUT "ENTER INCREMENTAL ANGLE. DEFAULT IS 5 DEGREES",Resolution
950 Resolution=INT(Resolution)
960 IF Resolution<1 THEN
970 PRINT "INCREMENTAL ANGLE MUST BE GREATER THAN OR EQUAL TO 1"
980 GOTO 930
990 ELSE
1000 END IF
1010 !
1020 DISP "DIGITIZE ORIGIN"
1030 OUTPUT 706;"SG"          ! SET TO SINGLE POINT MODE
1040 CALL Status              ! WAIT UNTIL GRAPHICS TABLET IS READY
1050 OUTPUT 706;"OD"          ! INSTRUCT GRAPHICS TABLET TO TRANSMIT POINT
1060 ENTER 706;Originx,Originy ! GET THE POINT FROM THE GRAPHICS TABLET
1070 OUTPUT 706;"BP24,125,4" ! INSTRUCT GRAPHICS TABLET TO BEEP
1080 !
1090 DISP "DIGITIZE ZERO DEGREE FULL SCALE"
1100 CALL Status
1110 OUTPUT 706;"OD"
1120 ENTER 706;Zerodegfsx,Zerodegfsy
1130 OUTPUT 706;"BP"
1140 !
1150 DISP "DIGITIZE MAXIMUM RESPONSE AXIS"
1160 CALL Status
1170 OUTPUT 706;"OD"
1180 ENTER 706;Mrax,Mray
1190 OUTPUT 706;"BP"
1200 !
1210 ! CALCULATE THE NUMBER OF DIGITIZING UNITS PER DECIBEL BY DIVIDING
1220 ! THE RADIUS (Rho) AT FULL SCALE BY THE NUMBER OF DECIBELS FULL SCALE
1230 Differencex=ABS(Zerodegfsx-Originx)
1240 Differencey=ABS(Zerodegfsy-Originy)
1250 CALL Calculateradius(Differencex,Differencey,Rho_zerodegfs)
1260 Gtabuperdb=Rho_zerodegfs/Dbfullscale
1270 !
1280 ! CALCULATE THE MAXIMUM RESPONSE AXIS VALUE BY DIVIDING THE VALUE FOR
1290 ! THE MAXIMUM RESPONSE AXIS IN DIGITIZING UNITS BY THE NUMBER OF
1300 ! DIGITIZING UNITS PER DECIBEL

```

```

1310 ! < ie. CONVERT FROM DIGITIZING UNITS TO USER UNITS >
1320 Differencex=ABS(Mrmax-Originx)
1330 Differencey=ABS(Mray-Originy)
1340 CALL Calculateradius(Differencex,Differencey,Mra)
1350 Mra=Mra/Gtabupendb
1360 !
1370 ! DETERMINE THE "SKEW ANGLE" TO CORRECT FOR ERRORS THAT WILL RESULT FROM
1380 ! THE ORIGINAL DIRECTIVITY PATTERN BEING POSITIONED CROOKED OR SKEWED ON
1390 ! THE GRAPHICS TABLET. THIS CORRECTION FACTOR ALLOWS THE USER TO PLACE THE
1400 ! ORIGINAL DIRECTIVITY PATTERN IN any POSITION inside THE GRAPHICS TABLET
1410 ! DIGITIZING AREA AND STILL GATHER ACCURATE DATA. IT ALSO RESULTS IN A
1420 ! TRUE POLAR CO-ORDINATE SYSTEM
1430 CALL Displacement(Zerodegfsx,Zerodegfsy,Xvalue,Yvalue)
1440 CALL Calcarctangent(Xvalue,Yvalue,Relativeangle)
1450 CALL Determinequad(Zerodegfsx,Zerodegfsy,Quadrant)
1460 CALL Convertto360(Relativeangle,Quadrant,Absoluteangle)
1470 Skew=Absoluteangle
1480 PLOTTER 7,6 IS OFF ! DEACTIVATE GRAPHICS TABLET
1490 ! < ie. DON'T SEND GRAPHICS COMMANDS TO IT >
1500 SUBEND
1510 !
1520 !
1530 !
1540 Entercurve: ! ENTER A DIRECTIVITY PATTERN USING THE 9111A GRAPHICS TABLET
1550 ! THE USER MUST START DIGITIZING FROM ZERO DEGREES AND
1560 ! CONTINUE UNTIL HE REACHES 360 DEGREES. 'Resolution'
1570 ! DETERMINES THE NUMBER OF DATA POINTS ACCEPTED.
1580 ! 360/'Resolution' SECTORS ARE CREATED AND A DATA POINT
1590 ! IS ENTERED FOR EACH SECTOR AND STORED IN ITS RESPECTIVE
1600 ! ARRAY ELEMENT
1610 SUB Entercurve
1620 OPTION BASE 0
1630 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mrmax,Mray
1640 COM Gtabupendb,Dbfullscale
1650 COM Radiusvalues(360),Angle(360)
1660 COM Mra
1670 COM Resolution
1680 COM Dirindex
1690 COM Skew
1700 DEG ! DEGREES MODE (NOT RADIANS)
1710 !
1720 ! "ENTERCURVE" EXPECTS TO HAVE A "COMMUNICATIONS CHANNEL" WITH THE
1730 ! GRAPHICS TABLET ALREADY ESTABLISHED. IT ALSO EXPECTS THE GRAPHICS
1740 ! TABLET TO HAVE BEEN INITIALIZED. < AS IN "SETUPAXES" >
1750 ! "ENTERCURVE WILL RE-ACTIVATE THE GRAPHICS TABLET IF IT HAS PREVIOUSLY
1760 ! BEEN DEACTIVATED
1770 !
1780 PLOTTER 7,6 IS ON
1790 OUTPUT 706;"CN" ! SET GRAPHICS TABLET TO CONTINUOUS
1800 ! SAMPLING MODE
1810 DISP "START DIGITIZING FROM ZERO DEGREES AND PROCEED COUNTER-CLOCKWISE"
1820 FOR Anglecounter=0 TO 360-Resolution STEP Resolution
1830 CALL Status ! WAIT UNTIL GRAPHICS TABLET IS READY
1840 OUTPUT 706;"OD"
1850 ENTER 706;Beamformx,Beamformy ! GET DATA POINT
1860 !
1870 ! CONVERT THE X AND Y COORDINATES TO POLAR COORDINATES BY CALLING THE
1880 ! FOLLOWING SUBPROGRAMS IN THE PROPER SEQUENCE
1890 !
1900 CALL Displacement(Beamformx,Beamformy,Xdisplacement,Ydisplacement)
1910 CALL Calculateradius(Xdisplacement,Ydisplacement,Radius)
1920 CALL Calcarctangent(Xdisplacement,Ydisplacement,Relativeangle)
1930 CALL Determinequad(Beamformx,Beamformy,Quadrant)
1940 CALL Convertto360(Relativeangle,Quadrant,Absoluteangle)
1950 !
1960 ! TEST TO SEE IF THE POINT LIES WITHIN THE SECTOR. IF IT DOES, THEN STORE

```

```

1970 ! IT IN ITS RESPECTIVE ARRAY ELEMENT. REMEMBER TO CORRECT FOR ANY "SKEW"
1980 ! ERRORS THAT MAY EXIST
1990 !
2000 IF (Absoluteangle-Skew>Anglecounter) AND (Absoluteangle-Skew<Anglecounter+
Resolution) THEN
2010 ELSE
2020 !
2030 ! IF 'Skew' IS NEGATIVE THEN PERFORM THE TEST BUT WITH 360 DEGREES ADDED
2040 ! TO MAKE THE ANGLE POSITIVE
2050 !
2060 IF (Absoluteangle+360-Skew>Anglecounter) AND (Absoluteangle+360-Skew<Angle
counter+Resolution) THEN
2070 Absoluteangle=Absoluteangle+360
2080 ELSE
2090 GOTO 1830
2100 END IF
2110 END IF
2120 DISP "ANGLE =";Absoluteangle-Skew,"db DOWN =";Mra-Radius
2130 Radiusvalues(Anglecounter)=Radius
2140 Angle(Anglecounter)=Absoluteangle-Skew
2150 BEEP
2160 NEXT Anglecounter
2170 !
2180 PLOTTER 7,6 IS OFF          ! DEACTIVATE GRAPHICS TABLET
2190 SUBEND
2200 !
2210 !
2220 !
2230 Displacement:      ! GIVEN A POINT'S X AND Y CO-ORDINATES, DETERMINE IT'S
2240                     ! RELATIVE X AND Y DISPLACEMENTS FROM THE ORIGIN
2250                     ! 'Xvalue' IS THE X-CO-ORDINATE OF THE POINT IN GRAPHIC
2260                     ! TABLET UNITS.
2270                     ! 'Yvalue' IS THE Y-CO-ORDINATE OF THE POINT IN GRAPHIC
2280                     ! TABLET UNITS
2290                     ! 'Xdisplacement' IS THE RELATIVE X-DISPLACEMENT FROM THE
2300                     ! ORIGIN IN USER DEFINED UNITS
2310                     ! 'Ydisplacement' IS THE RELATIVE Y-DISPLACEMENT FROM THE
2320                     ! ORIGIN IN USER DEFINED UNITS
2330 SUB Displacement(Xvalue,Yvalue,Xdisplacement,Ydisplacement)
2340 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mmax,Mray
2350 COM Gtabuperdb,Dbfullscale
2360 Xdisplacement=(Xvalue-Originx)/Gtabuperdb
2370 Ydisplacement=(Yvalue-Originy)/Gtabuperdb
2380 SUBEND
2390 !
2400 !
2410 !
2420 Calculateradius:   ! CALCULATE THE RADIUS USING THE PYTHAGAREAN THEOREM
2430                     ! (DISTANCE FORMULA)
2440 SUB Calculateradius(X,Y,R)
2450 R=SQR(X^2+Y^2)
2460 SUBEND
2470 !
2480 !
2490 !
2500 Calcarctangent:   ! CALCULATE THE VALUE OF THE ARCTANGENT GIVEN 'X' AND 'Y'
2510                     ! ALSO RESOLVE ANY INVALID INPUT PARAMETERS TO THE
2520                     ! ARCTANGENT FUNCTION BY SUBSTITUTING THE CORRECT VALUE
2530 SUB Calcarctangent(X,Y,Angle)
2540 DEG                ! DEGREES MODE (NOT RADIANS)
2550 ON ERROR GOTO Recovery
2560 Angle=ATN(Y/X)
2570 SUBEXIT
2580 Recovery:         ! ROUTINE TO RECOVER INVALID INPUT PARAMETERS TO
2590                     ! ARCTANGENT FUNCTION. (ie. DENOMINATOR CAN'T EQUAL
2600                     ! ZERO)

```

```

2610          ! ANGLE=ATN(Y/X) ; ASSUMES X DOES NOT 0
2620 IF ERRN=31 THEN ! IF DIVISION BY ZERO ERROR THEN CORRECT THE ERROR
2630          ! BY ASSIGNING THE CORRECT VALUE OR SOME ARBITRARY
2640          ! VALUE TO 'ANGLE'
2650 SELECT SGN(Y)   ! A NUMBER IS EITHER +,-, OR ZERO, SO DETERMINE THE
2660          ! CORRECT VALUE FOR THE 3 CASES OF 'Y'
2670 CASE 1          ! IF 'Y' IS POSITIVE THEN THE CORRECT ANSWER IS 90 DEGREES
2680          ! (ie. THE POINT LIES SOMEWHERE ON THE Y-AXIS, BUT ABOVE
2690          ! THE X-AXIS)
2700 Angle=0        ! 0 PRODUCES A VALUE OF 90 DEGREES AFTER BEING PROCESSED
2710          ! BY "CONVERTT0360"
2720 CASE 0          ! IF 'Y' IS ZERO THEN THE POINT IS AT THE ORIGIN, SO
2730          ! ARBITRARILY ASSIGN AN ANGLE OF ZERO DEGREES. THE VALUE
2740          ! OF THE RHO (RADIUS) WILL BE CALCULATED AS ZERO AND IT WILL
2750          ! BE OBVIOUS TO THE USER THAT THE POINT WAS AT THE ORIGIN
2760 Angle=451       ! 451 PRODUCES A VERY LARGE VALUE AFTER BEING PROCESSED BY
2770          ! "CONVERTT0360". THIS RESULTS IN "ENTERCURVE" REJECTING
2780          ! ANY POINTS THAT LIE AT THE ORIGIN BECAUSE THEY WILL BE
2790          ! OUT OF RANGE
2800 CASE -1        ! IF 'Y' IS NEGATIVE THEN THE CORRECT ANSWER IS 270 DEGREES
2810          ! (ie. THE POINT LIES SOMEWHERE ON THE Y-AXIS, BUT BELOW THE
2820          ! X-AXIS)
2830 Angle=0        ! 0 PRODUCES A VALUE OF 270 DEGREES AFTER BEING PROCESSED
2840          ! BY "CONVERTT0360"
2850 END SELECT     ! NO NEED FOR A 'CASE ELSE' STATEMENT AS A NUMBER IF EITHER
2860          ! +,- OR ZERO
2870 SUBEXIT        ! END OF ERROR HANDLER ROUTINE
2880 ELSE          ! IF THIS ISN'T A DIVISION BY ZERO ERROR, THEN DISPLAY THE
2890          ! ERROR MESSAGE AND PAUSE PROGRAM EXECUTION
2900 DISP ERRM$
2910 PAUSE
2920 END IF
2930 SUBEND
2940 !
2950 !
2960 !
2970 Determinequad: ! DETERMINE WHICH QUADRANT THE POINT LIES IN BASED UPON
2980          ! WHETHER THE POINT LIES TO THE LEFT OR RIGHT OF THE
2990          ! Y-AXIS AND ABOVE OR BELOW THE X-AXIS
3000 SUB Determinequad(Xvalue,Yvalue,Quadrant)
3010 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mmax,Mray
3020 IF Xvalue>Originx THEN          ! IN QUADRANT 1 OR 4 THEN
3030 IF Yvalue>Originy THEN          ! IN QUADRANT 1
3040 Quadrant=1
3050 ELSE
3060 Quadrant=4
3070 END IF
3080 ELSE          ! IN QUADRANT 2 OR 3 OR ON Y-AXIS THEN
3090 IF Yvalue>Originy THEN          ! IN QUADRANT 2
3100 Quadrant=2
3110 ELSE
3120 IF Xvalue=Originx THEN          ! IN QUADRANT 4 ( ON Y-AXIS )
3130 Quadrant=4
3140 ELSE
3150 Quadrant=3
3160 END IF
3170 END IF
3180 END IF
3190 SUBEND
3200 !
3210 !
3220 !
3230 Convertto360:   ! GIVEN THE PRINCIPAL ANGLE AND THE QUADRANT PRODUCE
3240          ! AN ANGULAR VALUE IN THE RANGE 0 TO 360 DEGREES
3250          ! 'Relativeangle' IS THE PRINCIPAL ANGLE
3260          ! 'Quadrant' IS IN THE RANGE 1-4 ( FROM CARTESIAN

```



```

3270             ! CO-ORDINATE SYSTEM >
3280             ! 'Absoluteangle' IS THE RESULTANT ANGULAR VALUE IN
3290             ! THE RANGE 0-360 DEGREES
3300 SUB Convertto360(Relativeangle,Quadrant,Absoluteangle)
3310 SELECT Quadrant
3320 CASE 1             ! FIRST QUADRANT      0 <= Relativeangle < 90
3330 Absoluteangle=Relativeangle+270
3340 CASE 2             ! SECOND QUADRANT    90 <= Relativeangle < 180
3350 Absoluteangle=Relativeangle+90
3360 CASE 3             ! THIRD QUADRANT    180 <= Relativeangle < 270
3370 Absoluteangle=Relativeangle+90
3380 CASE 4             ! FOURTH QUADRANT 270 <= Relativeangle < 360
3390 Absoluteangle=Relativeangle+270
3400 END SELECT
3410 SUBEND
3420 !
3430 !
3440 !
3450 Clearscreen: ! CLEAR THE ALPHA RASTER BY PRINTING 24 BLANK LINES
3460 SUB Clearscreen
3470 FOR Line=1 TO 24
3480 PRINT
3490 NEXT Line
3500 SUBEND
3510 !
3520 !
3530 !
3540 Displaypattern: ! DRAW THE DIRECTIVITY PATTERN ON THE GRAPHICS RASTER
3550             ! AND LABEL THE DIRECTIVITY INDEX
3560 SUB Displaypattern
3570 OPTION BASE 0
3580 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mrax,Mray
3590 COM Gtabuperdb,Dbfullscale
3600 COM Radiusvalues(360),Angle(360)
3610 COM Mra
3620 COM Resolution
3630 COM Dirindex
3640 DEG             ! DEGREES MODE ( NOT RADIAN )
3650 PLOTTER IS 13,"GRAPHICS" ! ACTIVATE THE CRT GRAPHICS RASTER
3660 GRAPHICS             ! ENABLE GRAPHICS
3670 GCLEAR             ! CLEAR GRAPHICS RASTER
3680 LOCATE 20,95,0,100
3690 MOVE 37,2
3700 CSIZE 3.2
3710 LABEL "Directivity index =";DROUND(Dirindex,6)
3720 SCALE -Dbfullscale*.75,Dbfullscale*.75,-Dbfullscale,Dbfullscale
3730 FRAME
3740 AXES 10,10,0,0,1,1
3750 FOR Anglecounter=0 TO 360-Resolution STEP Resolution
3760 X=Radiusvalues(Anglecounter)*COS(Angle(Anglecounter)+90)
3770 Y=Radiusvalues(Anglecounter)*SIN(Angle(Anglecounter)+90)
3780 IF Anglecounter=0 THEN
3790 MOVE X,Y
3800 ELSE
3810 DRAW X,Y
3820 END IF
3830 NEXT Anglecounter
3840 ! COMPLETE THE DIRECTIVITY PATTERN PLOT BY DRAWING A LINE FROM THE
3850 ! LAST POINT TO THE FIRST POINT.
3860 DRAW Radiusvalues(0)*COS(Angle(0)+90),Radiusvalues(0)*SIN(Angle(0)+90)
3870 PLOTTER 13 IS OFF             ! DEACTIVATE THE CRT GRAPHICS RASTER
3880 SUBEND
3890 !
3900 !
3910 !
3920 Calcdirindex: ! CALCULATE THE DIRECTIVITY INDEX GIVEN THE RADIUS (Rho)

```

```

3930          ! AND THE ANGLE ( 0-360 DEGREES )
3940 SUB Calcdirindex
3950 OPTION BASE 0
3960 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mmax,Mray
3970 COM Gtabuperdb,Dbfullscale
3980 COM Radiusvalues(360),Angle(360)
3990 COM Mra
4000 COM Resolution
4010 COM Dirindex
4020 DEG          ! DEGREES MODE, NOT RADIANS
4030 Sum=0
4040 Anglecounter=0
4050 FOR Anglecounter=0 TO 360-Resolution STEP Resolution
4060 Temp=ABS(SIN(Angle(Anglecounter)))*10^(-(Mra-Radiusvalues(Anglecounter))/1
0)
4070 Sum=Temp+Sum
4080 NEXT Anglecounter
4090 Dirfactor=229.2/(Resolution*Sum)
4100 Dirindex=10*LGT(Dirfactor)
4110 SUBEND
4120 !
4130 !
4140 !
4150 Printcopy:   ! DUMP THE CONTENTS OF THE GRAPHICS RASTER TO THE PRINTER
4160              ! ONLY THE DIRECTIVITY PATTERN AND ITS RESPECTIVE LABELS WILL
4170              ! APPEAR ON THE PRINTER OUTPUT
4180 SUB Printcopy
4190 Yesno$="N"
4200 INPUT "DUMP DIRECTIVITY PATTERN TO PRINTER - Y/N. DEFAULT IS 'N'",Yesno$
4210 IF Yesno$="Y" THEN
4220 PLOTTER 13 IS ON
4230 PRINTER IS 0                      ! OUTPUT DEVICE = PRINTER
4240 DUMP GRAPHICS
4250 PRINT CHR$(12)                    ! ADVANCE TO TOP OF NEXT SHEET OF PAPER
4260 PRINTER IS 16                     ! OUTPUT DEVICE = CRT
4270 PLOTTER 13 IS OFF
4280 ELSE
4290 END IF
4300 SUBEND
4310 !
4320 !
4330 !
4340 Status:      ! INTERROGATE THE 9111A GRAPHICS TABLET TO SEE IF IT'S READY
4350              ! READ THE STATUS WORD AND TEST BIT 2. THE GRAPHICS TABLET IS
4360              ! READY WHEN BIT 2 = 1
4370 SUB Status
4380 STATUS 7,6;S
4390 IF BIT(S,2)=0 THEN 4380
4400 SUBEND
4410 !
4420 !
4430 !
4440 Plotcopy:    ! PLOT THE DIRECTIVITY PATTERN ON THE PLOTTER AND LABEL THE
4450              ! DIRECTIVITY INDEX
4460 SUB Plotcopy
4470 OPTION BASE 0
4480 COM Originx,Originy,Zerodegfsx,Zerodegfsy,Mmax,Mray
4490 COM Gtabuperdb,Dbfullscale
4500 COM Radiusvalues(360),Angle(360)
4510 COM Mra
4520 COM Resolution
4530 COM Dirindex
4540 DEG          ! DEGREES MODE (NOT RADIANS)
4550 Yesno$="N"
4560 Dimflag=0
4570 INPUT "DUMP DIRECTIVITY PATTERN TO PLOTTER - Y/N. DEFAULT IS 'N'",Yesno$

```

```


4580 IF Yesno$="Y" THEN
4590 IF Dimflag=1 THEN 4630
4600 DIM Temp$(2)[35]
4610 Dimflag=1
4620 PLOTTER IS 7,5,"9872A"
4630 Temp$(0)=" "
4640 Temp$(1)=" "
4650 Temp$(2)=" "
4660 INPUT "ENTER LABEL FOR FIRST LINE",Temp$(0)
4670 INPUT "ENTER LABEL FOR SECOND LINE",Temp$(1)
4680 INPUT "ENTER DATE (OPTIONAL)",Temp$(2)
4690 LOCATE 0,RATIO*100,0,100
4700 FRAME
4710 LOG 2
4720 LDIR 90
4730 CSIZE 2.6
4740 MOVE 120,4 ! DRAW LABEL 1
4750 LABEL Temp$(0)
4760 MOVE 125,4 ! DRAW LABEL 2
4770 LABEL Temp$(1)
4780 MOVE 130,4 ! DRAW LABEL 3
4790 LABEL Temp$(2)
4800 MOVE 130,53 ! LABEL DIRECTIVITY INDEX
4810 LABEL "Directivity index =";DROUND(Dirindex,6)
4820 LDIR 0
4830 ! SCALE PLOTTING AREA TO USER DEFINED UNITS
4840 SCALE -Dbfullscale,Dbfullscale,-Dbfullscale*.75,Dbfullscale*.75
4850 ! DRAW AXES WITH TICK MARKS EVERY 10 db
4860 AXES 10,10,0,0,1,1
4870 ! DRAW DIRECTIVITY PATTERN
4880 FOR Anglecounter=0 TO 360-Resolution STEP Resolution
4890 X=Radiusvalues(Anglecounter)*COS(Angle(Anglecounter)+180)
4900 Y=Radiusvalues(Anglecounter)*SIN(Angle(Anglecounter)+180)
4910 IF Anglecounter=0 THEN
4920 MOVE X,Y
4930 ELSE
4940 DRAW X,Y
4950 END IF
4960 NEXT Anglecounter
4970 ! COMPLETE THE DIRECTIVITY PATTERN PLOT BY DRAWING A LINE FROM THE
4980 ! LAST POINT TO THE FIRST POINT.
4990 DRAW Radiusvalues(0)*COS(Angle(0)+180),Radiusvalues(0)*SIN(Angle(0)+180)
5000 PEN 0
5010 Yesno$="N"
5020 INPUT "MAKE ANOTHER PLOT - Y/N. DEFAULT IS (N)0",Yesno$
5030 IF Yesno$="Y" THEN 4570
5040 PLOTTER 7,5 IS OFF
5050 SUBEXIT
5060 ELSE
5070 SUBEXIT
5080 END IF
5090 SUBEND

```


USING THE PROGRAM


This program is very easy to use. Most input responses require simple "yes" or "no" answers. The process of digitizing a directivity pattern consumes the most time, and this process uses the stylus of the graphics tablet for input. To execute the program, the operator must complete the following simple steps:

1. Turn on the HP 9845 computer
2. Turn on the HP 9872A plotter
3. Turn on the HP 9111A graphics tablet
4. Load "CALCDI" from the proper mass storage device
5. Press the "RUN" key
6. Respond accordingly to computer generated questions





Sample circle pattern





Sample wedge pattern generated from
previous page







Sample spike pattern generated from
previous page

Original Pattern


DI = 18.419

SCALE: 1dB per small
radial division





Sample of typical pattern generated
from previous page using 5° resolution



Sample of typical pattern generated
from previous page using 2° resolution